
Abbreviations 

PAS  Pilz Automation Suite (Software-Platform) 

PSS  Programmable Control System (DE: Programmierbares Steuerungssystem) 

POU  Program Organisation Unit 

SSM  Safe Speed Monitoring 

SDI-M  Safe Direction Monitoring 

SOS-M  Safe Operating Stop Monitoring 

SSR-M  Safe Speed Range Monitoring 

SS1  Safe Stop 1 

STO  Safe Torque Off 

1. Useful documentation 

Reading the documentation listed below is necessary for understanding this application note. 

The availability of the indicated tools and safe handling are also presupposed with the user. 

1.1. Documentation from Pilz GmbH & Co. KG 

No. Description Item No. 

1 Pilz international homepage, download section www.pilz.com 

2 Operating Manual PSSu H PLC1 FS SN SD 21939-EN-xx 

3 Operating Manual PSSu E F 4DI 21310-EN-xx 

4 Operating Manual PSSu E S 4DI(-T) 21340-EN-xx 

5 Operating Manual PSSu E S 4DO 0.5 21346-EN-xx 

6 Operating Manual PSSu K F EI 1003303-EN-xx 

7 Operating Manual PSEN me2, PSEN me3 1003287-2EN-xx 

8 Operating Manual PMCprotego D.48, D.72 1001735-EN-xx 

9 Operating Manual PMCtendo SZ 1002405-EN-xx 

10 System Description Automation system PSS 4000 1001467-EN-xx 

11 Safety Manual Automation system PSS 4000 1001468-EN-xx 

12 PAS4000 Online Help - 

13 PASmotion Online Help - 

1.2. Documentation from other sources of information 

No. Description Item No. 

1   

2   

2. Hardware configuration 

2.1. Pilz products 

No. Description Order number Version Number 

1 PSSu H PLC1 FS SN SD 312070 FW 1.12.0 1 

2 PSSu E F 4DI 312200 - 2 

3 PSSu E S 4DI 312400 - 1 

4 PSSu E S 4DO 0.5 312405 - 1 

5 PSSu K F EI 312433 - 1 

6 PSSu A Con Set1 C 313114 - 1 

7 PSSu BP 1/8C 312601 - 4 

8 PSEN me2 / 2AS 570200 - 2 

9 PMCprotego D.72/000/0/0/2/208-480V AC 8176426-1 - 1 

10 PMCtendo SZ.75/0/2/2/7/F/H/30/00 8177026 - 1 

11 Brake resistor 1600W/10R/T/U 8176364 - 1 

12 PMCcable FD/D4B6/005/Q25/S1 (Feedback cable EnDat) 8177101 - 1 

13 PMCcable M2/C1B3/005/6Q0/S1 (Power cable) 8177211 - 1 

14 PNOZ msi b4 Box 773845 - 1 

15 MM A MINI-IO-CAB01 1.5M 772200 - 1 

16 PAS4000 - V1.12.0 1 

17 PASmotion - V1.0.1 1 

2.2. Hardware configuration 

 

Fig. 1: Hardware configuration 

This example illustrates safely monitored speed (rotational speed) and safely monitored 

direction of rotation with a PSS 4000 PLC and the module PSSu K F EI. 

A safe rotary encoder (B1) is used to record the movement of the motor. The rotary encoder is 

connected to the module PSSu K F EI (A1), which evaluates the signals safely. 

Two PSEN me2 (S1, S2) are used to monitor a safety gate. 

The motor (M1) is connected to a servo amplifier PMCprotego D (A2). Several Pilz function 

blocks (FS-FBs) from the PAS4000 Library are used in the PLC program that implements the 

application: 

 

 

 FS_SafetyGate    Safety Gate Monitoring 

 FS_EI_SSM0 

 FS_EI_SDIM 

 FS_EI_SOSM    Communication with the Module PSSu K F EI 

 FS_EI_SSM1_SSRM 

 FS_EI_Basic 

Fig. 2: PSS 4000 Configuration 

The module PSSu K F EI is configured as follows: 

 

Fig. 3: Configuration of the module PSSu K F EI 

Guidelines for the sections EI (X31, X32) and CIO (X13:9) 

 

Sensor type: 

The rotary encoder used in this case supplies a Sin/Cos signal 1 Vss. 

 

Increments/revolution for user-defined units: 

The rotary encoder supplies 2048 increments per revolution. This value is needed for 

conversion into other physical sizes. 

 

Max. frequency AB: 

The data sheet for the rotary encoder lists a maximum frequency of 200 kHz on tracks A and B. 

Entries are made in mHz. The module PSSu K F EI monitors whether this value is exceeded. 

 

Sensor resolution: 

Default value, as only relevant on sensors with track Z. 

 

Max. frequency Z: 

Default value, as only relevant on sensors with track Z. 

 

Validation cut-off frequency: 

Default value, as only relevant when using proximity switches for tracks A/B. 

 

Monitor track S / Umax track S / Umin track S: 

Track S is not monitored in this application. 

 

Invert direction of movement: 

Due to the mounting, in this example the direction of movement is inverted internally within the 

module, in order to monitor the physically correct direction of movement. 

 

Hysteresis: 

The stated value of 2% ensures that once the monitoring function SSM1 has reacted, the 

enable will not be reset until the speed value is 2% below the configured limit value. 

 

Enable cascading: 

Cascading is not possible in this example because only one module PSSu K F EI is used. 

 

 

The configuration of the servo amplifier is not described in this Application Note. 

3. Application Task 

3.1. Functional description 

3.1.1. Plant overview 

This application example uses a steel strip finishing plant to show how speed (rotational speed) 

and direction of rotation can be monitored safely using a PSS 4000 PLC and the module PSSu 

K F EI. 

 

A steel strip is unwound from a reel and transported onwards via rollers for processing. 

 

Fig. 4: Steel strip processing application 

In automatic mode, safety fences and safety gates protect staff from hazards arising from the 

transportation of materials. 

 

The risk analysis indicated that interventions to maintain operation are only permitted at 

reduced material speed and with the correct transport direction, i.e. if a safety gate is opened, 

the plant must be safely monitored for reduced speed. 

 

The material speed can only be recorded indirectly via the speed of the reel. As a result, the 

limit value for monitoring the reduced speed should also be set as the rotational speed value. 

 

As the diameter of the coil changes as it unwinds and the material speed is constant, the speed 

of the reel will change. As a result, the limit value for monitoring the reduced speed must be 

adjusted, based on the coil diameter. 

The coil diameter is recorded safely using the sensors; from this, signals are derived for 

switching the limit value in 3 steps. The way in which the diameter is recorded safely is not 

described in this Application Note. 

 

Plant details (example):  

Diameter of coil:  500 … 1500 mm 

Processing speed:  max. 500 m/min 

Reduced speed:  max. 30 m/min 

 

Calculation of limit values for three diameter ranges: 

 

The following applies:  Vu = d x π x n 

where:    Vu : Circumferential speed 

    d   : Diameter 

    n   : Speed 

 

Fig. 5: Speeds and limit values based on coil diameter 

A max. permitted reduced speed of 30 m/min produces the following speeds and limit values: 

Diameter of – to [mm] Speed of – to [rpm] Speed limit value [rpm] 

1500 - 1101 6 – 8 6 

1100 - 701 8 – 13 8 

700 - 500 13 - 19 13 

 

When the limit values are observed, the conveyor speed is reduced to approximately 20 m/min 

as the diameter decreases. 

 

 

3.1.2. Safety gate monitoring 

The block FS_SafetyGate is used to monitor a safety gate. 

 

The enable signal for safety gate monitoring is used to activate and deactivate the monitoring 

function SSM1. If the enable is missing (safety gate open or fault on the safety gate), the 

monitoring function SSM1 is activated, i.e. compliance with the reduced speed is monitored. 

 

Note: "Activate a monitoring function" means that the result of the monitoring function affects the 

outputs on the module PSSu K F EI. If the monitoring function reacts, outputs O0, O1 and O2 

are shut down. 

3.1.3. Monitoring rotational speed/speed 

The monitoring function SSM1 (Safe Speed Monitoring 1) of the module PSSu K F EI is used to 

monitor reduced speed when the safety gate is open. 

 

The monitoring function SSM1 is configured via an instance of the Pilz function block 

FS_EI_SSM1_SSRM. 

 

The speed limit value is adjusted based on the coil diameter. For this purpose, the respective 

diameter range is detected safely (not described in this Application Note) and is supplied via 

three SAFEBOOL variables. When switching to another diameter range, a pulse is formed by a 

task cycle and is used to start the transfer of the respective limit value. Simultaneously a timer is 

started, which is used to monitor the parameter transfer. 

 

As the parameters are being transferred, the output signals SSM1Enable, SSRMEnable and 

ParameterTransmissionReady on the block FS_EI_SSM1_SSRM are set to FALSE. The 

successful transfer of the limit values is displayed on the output ParameterTransmissionReady. 

In the event of an error (ParameterTransmissionReady = FALSE after time monitoring has 

elapsed), the module's outputs O0, O1 and O2 are shut down. The output signals SSM1Enable 

and ParameterTransmissionReady are muted while the timer is running. 

 

The monitoring functions SSM0, SSR-M, SOS-M are deactivated via instances of the Pilz 

function blocks FS_EI_SSM0, FS_EI_SOSM, FS_EI_SSM1_SSRM. Without deactivation, it 

would not be possible to put the plant into operation due to one or more missing enables.  

 

  



Application Task 

Application Note PSS 4000 - Rotational speed monitoring with safe encoder 13 
1003746_EN_02 

It is possible to switch between the limit values without interrupting operation, provided the limit 

values are not violated when switching. 

 

A value of 50 ms is set for monitoring the time of the switching process. With the present 

hardware structure and a task cycle time of 10 ms, the transfer of limit values is normally always 

completed within this time. 

3.1.4. Monitoring rotational speed/direction of movement 

The monitoring function SDI-M (Safe Direction Monitoring) of the module PSSu K F EI is used 

for permanent monitoring of the direction of movement. 

 

The monitoring function SDI-M is configured via an instance of the Pilz function block 

FS_EI_SDIM. 

3.1.5. Triggering of safety function SS1 

SS1 (Safe Stop 1) is implemented using the outputs O0, O1 and O2 of the module PSSu K F EI 

and the servo amplifier PMCProtego D. 

 

Should a monitoring function react on the module PSSu K F EI or outputs O0, O1 and O2 be 

shut down via the block inputs SetO0, SetO1 and SetO2, first the servo amplifier PMCprotego D 

will be disabled via the output O0 (ENABLE input = FALSE). The servo amplifier brings the 

motor to 0 speed using the internal emergency braking ramp. 

 

Then, after the configured delay time has elapsed, the safety function STO is triggered in the 

servo amplifier PMCprotego D via the outputs O1/O2 (inputs STO1-ENABLE = FALSE and 

STO2-ENABLE = FALSE). 

3.1.6. Basic functions, error acknowledgement and restart interlock 

Should a monitoring function react on the module PSSu K F EI, causing the outputs O0, O1 and 

O2 to shut down, or should the module detect an error, a restart interlock is set on the module. 

 

The restart interlock is reset via an instance of the block FS_EI_Basic. Further functions of the 

block are: 

 

 Driving the outputs O0, O1, O2 

 Error acknowledgement 

 Output of rotational speed/speed value (current value) 

 Output of status and diagnostic messages 

 Cascading (not used in this application) 

 

In this example, separate control elements are provided to reset the restart interlock and for the 

error acknowledgement. 

A signal is connected to the inputs for activating outputs O0, O1 and O; this is composed as 

follows: 

 

 Time monitoring active after change in limit value or parameter transfer complete 

 

Outputs O0, O1 and O2 are shut down if a parameter transfer has not been completed. 

 

When switching the servo amplifier PMCprotego D back on it is important to note that the 

enable signal ENABLE is set to TRUE no sooner than 100 ms after the signals STO1-ENABLE 

and STO2-ENABLE. As a result, the control signal for O0 and the reset of the restart interlock 

are delayed for output O0 by 200 ms via a time element. 

 

3.1.7. Displays and interlock conditions 

In this example, the following errors are signalled via indicator lights: 

 

 No enable of SSM1 while reduced speed is monitored 

(when the safety gate is open) 

 No enable of SDIM (positive direction) 

 

Additional messages are signalled via the PAS4000 diagnostic system. 

3.2. Configuration of the monitoring functions 

A wizard is available in PAS4000 for configuration of the monitoring functions on the module 

PSSu K F EI. It is called up via the "Monitoring functions" button, which appears in the 

properties window when selecting the module in the hardware configuration. 

 

Fig. 6: Configuration of the monitoring functions: User-defined units 

The user-defined units are set on the first tab.  

 

Mechanical conversion: Between the motor and plant there is a gear with a transmission ratio of 

10:1. As a result, a ratio of 1:10 is entered for scaling (1 revolution on the load corresponds to 

10 revolutions on the motor). 

 

Time conversion: As the speed of the plant is stated in m/min, the rotational speed is converted 

into revolutions/min (1 min corresponds to 60 s). 

 

Units: "Increments" are selected as the position unit, "RPM" is selected as the speed unit. 

 

The result of the calculations can be seen in Fig. 6 under "Information". 

 

The unit calculator is not used in this example. 

 

The monitoring function SSM0 is configured on the second tab. The default value 0 for 

maximum speed is not changed, as the monitoring function in the PLC program is deactivated. 

 

  



Application Task 

Application Note PSS 4000 - Rotational speed monitoring with safe encoder 16 
1003746_EN_02 

 

Fig. 7: Configuration of the monitoring functions: SSM0 

No entries are made on the tabs SSM1 and SSR-M, as the limit value for safe speed monitoring 

is specified via the PLC program. 

 

Fig. 8: Configuration of the monitoring functions: SSM1   

Fig. 9: Configuration of the monitoring functions: SSR-M 

On the SDI-M tab, 100 increments are entered as tolerance values for safe direction monitoring 

in both the positive and negative direction. 

 

Fig. 10: Configuration of the monitoring functions: SDI-M   

As the monitoring function SOS-M is not used in this application, the default values can be 

adopted on the tab SOS-M. 

 

Fig. 11: Configuration of the monitoring functions: SOS-M 

The drive of the outputs O0, O1 and O2 is configured on the last tab: 

 

 The default setting is adopted, where the outputs are switched depending on the monitoring 

result (figures [1], [2], [5]). 

 Automatic STO for outputs O1 and O2 is not activated (figures [3], [6]). 

 An STO delay time of 1000 ms is specified for outputs O1 and O2 (figures [4], [7]). 

 The standstill limit value for automatic STO is irrelevant in this example (figure [8]). 

Fig. 12: Configuration of the monitoring functions: SS1-M / STO 

3.3. Safety assessment 

3.3.1. Safety assessment for safety gate monitoring 

The safety control system monitors the input circuits on the 2-channel safety gate monitoring 

device for short circuit and shorts between contacts. The contacts are provided with test pulses 

for this purpose. 

 

The function block FS_SafetyGate monitors the switching of the safety gate contacts for 

simultaneity. A reset is required after the safety gate is closed. 

3.3.2. Safety assessment for rotary encoder and module PSSu K F EI 

An error in the PSSu K F EI module does not lead to the loss of the safety function. 

 

Open circuits on any individual track of the rotary encoder are detected by the module PSSu K 

F EI and lead to a shutdown of the outputs O0, O1 and O2. 

 

Outputs O1 and O2 are monitored by the module PSSu K F EI through cyclical output tests. If a 

defective output or a short circuit is detected, outputs O0, O1 and O2 are shut down. 

 

Errors that occur when transferring limit values are detected by the block FS_EI_SSM1_SSRM 

and lead to a shutdown of the outputs O0, O1 and O2. Depending on the system structure it can 

take up to 500 ms to transfer parameters (monitoring time of block FS_EI_SSM1_SSRM to error 

message "DiagParameterError"). In this application the parameters are transferred more 

quickly, so that any error when transferring parameters will result in a shutdown just 50 ms after 

the parameter transfer is started. 

 

The diameter-dependent speed limit values for the monitored reduced speed were determined 

before the PLC program was created and were stored in the program as a constant.  As a 

result, a parameter error due to a miscalculation at runtime is excluded. A parameter value 

outside the permitted value range is detected by the module PSSu K F EI. 

 

Forced dynamisation must be performed: The safe sensor for speed detection (B1) must be 

moved so that a signal changes on all the connected tracks within an 8 hour period. 

 

Common cause failure (CCF), which can cause the encoder system to fail, e.g. due to a 

shearpin breakage and slippage, must be excluded by suitable (e.g. design) measures or will 

require further measures for detection. 

3.3.3. Safety assessment for motor control 

An error in the PMCprotego D does not lead to the loss of the safety function. 

Hazardous situations must not be allowed to arise due to SS1 braking ramps and stopping 

times. 

The best possible mechanics at the drive (overdimensioning) are assumed (fault exclusion for 

broken shearpin). 

The operator must ensure that the function of the safe pulse disabler is tested periodically, after 

8 hours at the latest, by triggering safety functions SS1 or STO: 

 By restarting after safety functions SS1 or STO have been triggered as a condition of 

operation 

 By restarting after safety function SS1 has been triggered by the operator. ---------- 

 

3.4. Functional safety 

3.4.1. Safety-related characteristics in accordance with EN ISO 13849-1 

No. Safety function PL Safety-related parts of the control 

system 

1 The plant is stopped if the permitted reduced 

speed is exceeded while the safety gate is 

open. (SSM) 

PL d Input (Safe encoder) 

Input (PSENme 2/2AS) 

Input (PSSu K F EI) 

Logic (PSSu E F 4DI) 

Logic (PSSu K F EI) 

Logic (PSSu H PLC1 FS SN) 

Output (PSSu K F EI) 

Actuator (PMCprotego D) 

2 The plant is stopped if the plant's direction of 

movement is outside the permitted tolerance. 

(SDI) 

PL d Input (safe encoder) 

Input (PSSu K F EI) 

Logic (PSSu K F EI) 

Logic (PSSu H PLC1 FS SN) 

Output (PSSu K F EI) 

Actuator (PMCprotego D) 

 

Prerequisites 

No. Description Identification 

1 Common cause failure (CCF): Requirements are considered to be met 

(must be tested on implementation) 

2 Mission time: general 20 years 

Sensor B1 10 years 

3 Characteristic data of the sensor B1 PFHd 4,345 *10
-9
   1/h 

PL e 

4 Operating interval (electromechanical 

components): 

Sensor 1 operation all 2 hours for safety gate 

 

Please note the further requirements of EN ISO 13849-1, e.g. requirements for avoiding  

systematic faults. 

 

CAUTION 

The calculation of the performance level is only valid when used with a PMCtendo motor. 

3.4.2. Safety-related characteristics in accordance with EN 62061 

No. Safety-related control function (SRCF) Safety Integrity 

Level 

Subsystems 

1 The plant is stopped if the permitted reduced 

speed is exceeded while the safety gate is 

open. (SSM) 

SIL 2 Input (safe encoder) 

Input (PSENme 2/2AS) 

Input (PSSu K F EI) 

Logic (PSSu E F 4DI) 

Logic (PSSu K F EI) 

Logic (PSSu H PLC1 FS SN) 

Output (PSSu K F EI) 

Actuator (PMCprotego D) 

2 The plant is stopped if the plant's direction of 

movement is outside the permitted tolerance. 

(SDI) 

SIL 2 Input (safe encoder) 

Input (PSSu K F EI) 

Logic (PSSu K F EI) 

Logic (PSSu H PLC1 FS SN) 

Output (PSSu K F EI) 

Actuator (PMCprotego D) 

 

Prerequisites 

No. Description Identification 

1 Common cause failure (CCF): ß = 2 % (must be tested on 

implementation) 

2 Proof-Test Intervall: general 20 years 

Sensor B1 10 years 

3 Characteristic data of the sensor B1 PFHd 4,345 *10
-9
   1/h 

SIL 3 

4 Operating interval (electromechanical 

components): 

Sensor 1 operation all 2 hours for safety gate 

 

Please note the further requirements of EN 62061, e.g. requirements for systematic safety 

integrity. 

 

CAUTION 

The calculation of the safety integrity level is only valid when used with a PMCtendo motor. 

3.5. PAS project 

3.5.1. Program in the FS resource 

3.5.1.1. Declaration part  

PROGRAM MAIN_FS 

VAR 

//--Diameter ranges and switching of limit value---------------------------- 

 DiameterRange1    : SAFEBOOL; 

 DiameterRange2    : SAFEBOOL; 

 DiameterRange3    : SAFEBOOL; 

 SpeedRange1_RTrig   : R_TRIG; 

 SpeedRange2_RTrig   : R_TRIG; 

 SpeedRange3_RTrig   : R_TRIG; 

 SpeedRange1_REdge   : SAFEBOOL; 

 SpeedRange2_REdge   : SAFEBOOL; 

 SpeedRange3_REdge   : SAFEBOOL; 

 TimerChangeSpeedRange   : TP; 

 StartTimerChangeSpeedRange  : SAFEBOOL; 

 TimerChangeSpeedRangeRunning  : SAFEBOOL; 

 

//--Safety Gate-------------------------------------------------------------- 

 SafetyGate    : FS_SafetyGate; 

 SafetyGateEnable   : SAFEBOOL; 

 

//--Enable for set the outputs of the module PSSu K F EI--------------------- 

 EnableOutputs    : SAFEBOOL; 

 TimerEnableServo   : TON; 

 

//-- Speed monitoring functions PSSu K F EI --------------------------------- 
 KFEI_SSM0    : FS_EI_SSM0; 

 SSM0Enable    : SAFEBOOL; 

 

 KFEI_SOSM    : FS_EI_SOSM; 

 SOSMSet     : SAFEBOOL; 

 SOSMEnable    : SAFEBOOL; 

 

 KFEI_SDIM    : FS_EI_SDIM; 

 SDIMSet     AT %I* : SAFEBOOL; 

 PosEnable    : SAFEBOOL; 

 NegEnable    : SAFEBOOL; 

 

 KFEI_SSM1_SSRM    : FS_EI_SSM1_SSRM; 

 MaximumSpeed    : SAFEUDINT; 

 SSM1Enable    : SAFEBOOL; 

 SSRMEnable    : SAFEBOOL; 

 ParameterTransmissionReady  : SAFEBOOL; 

 DiagParameterError   : SAFEBOOL; 

 

 KFEI_Basic    : FS_EI_Basic; 

 ResetRestartInterlock   AT %I* : SAFEBOOL; 

 ResetError    AT %I* : BOOL; 

 SpeedValid     : SAFEBOOL; 

 SpeedValue    : SAFEUDINT; 

 StateCascading    : BOOL; 

 StateSTO_Timer1    : SAFEBOOL; 

 StateSTO_Timer2    : SAFEBOOL; 

 AutomaticSTO     : SAFEBOOL; 

 DiagRestartInterlock    : SAFEBOOL; 

 DiagEncoderMissing    : BOOL; 

 DiagModuleError    : BOOL; 

  

//--Messages----------------------------------------------------------------- 

 SSM1Error    AT %Q* : BOOL; 

 SDIMError    AT %Q* : BOOL; 

 SSM1Error_SR    : SR; 

 SDIMError_SR    : SR; 

END_VAR 

VAR CONSTANT 

//--Speed limit values ------------------------------------------------------ 

 MaxSpeed1    : SAFEUDINT := UDINT#6;  // rpm 

 MaxSpeed2    : SAFEUDINT := UDINT#8;  // rpm 

 MaxSpeed3    : SAFEUDINT := UDINT#13; // rpm 

END_VAR 

 

3.5.1.2. Instruction part 

//-- Safety Gate ----------------------------------------------------------- 

SafetyGate( 

 SwitchType                 := USINT#3, 

 AutoStart                  := TRUE, 

 AutoReset                  := FALSE, 

 MonitoredReset             := TRUE, 

 StartupTest                := FALSE, 

 SimultaneityTime           := T#1000ms, 

 DelayTime                  := T#30ms, 

 Enable                     => SafetyGateEnable 

 ); 

 

//-- Recording of diameter change ------------------------------------------- 
// Pulse edge evaluation of safe signals for recording diameter; so: 

// - Pulse for change of speed limit value (FS_EI_SS1M_SSRM) 

// - Timer is started (50 ms) to monitor the transfer of parameters 

SpeedRange1_RTrig( 

CLK := DiameterRange1 AND NOT DiameterRange2 AND NOT DiameterRange3 

       AND SpeedValid, 

Q   => SpeedRange1_REdge 

); 

SpeedRange2_RTrig( 

 CLK := DiameterRange2 AND NOT DiameterRange1 AND NOT DiameterRange3 

        AND SpeedValid, 

 Q   => SpeedRange2_REdge 

 ); 

SpeedRange3_RTrig( 

 CLK := DiameterRange3 AND NOT DiameterRange1 AND NOT DiameterRange2 

        AND SpeedValid, 

 Q   => SpeedRange3_REdge 

 ); 

 

StartTimerChangeSpeedRange := SpeedRange1_REdge OR SpeedRange2_REdge OR 

                              SpeedRange3_REdge; 

 

TimerChangeSpeedRange( 

 IN := StartTimerChangeSpeedRange, 

 PT := T#50ms, 

 Q  => TimerChangeSpeedRangeRunning 

 ); 

 

//-- Enable for setting the outputs of the module PSSu K F EI --------------- 
// Parameter transfer is running (monitoring timer is running) or is complete. 

EnableOutputs := TimerChangeSpeedRangeRunning OR ParameterTransmissionReady; 

 

//-- Selection of speed limit values ---------------------------------------- 
IF SpeedRange1_REdge THEN   // Diameter range 1 

 MaximumSpeed        := MaxSpeed1; // MaximumSpeed limit value for SSM1 

  

ELSIF SpeedRange2_REdge THEN   // Diameter range 2 

 MaximumSpeed        := MaxSpeed2; // MaximumSpeed limit value for SSM1 

 

 ELSIF SpeedRange3_REdge THEN  // Diameter range 3 

  MaximumSpeed := MaxSpeed3; // MaximumSpeed limit value for SSM1 

 

ELSE 

 MaximumSpeed        := MaximumSpeed; // Limit value unchanged; 

 

END_IF; 

//-- Monitoring functions PSSu K F EI --------------------------------------- 
// Safe Speed Monitoring 1 and Safe Speed Range Monitoring 

KFEI_SSM1_SSRM( 

 SSM1DeactivateSS1M         := SafetyGateEnable, // SSM1 activated, if S-Gate open 

 MaximumSpeed               := MaximumSpeed, 

 SSRMDeactivateSS1M         := TRUE,      // SSRM deactivated 

 MinimumSpeed               := UDINT#0, 

 SetParameter               := StartTimerChangeSpeedRange, 

 SSM1Enable                 => SSM1Enable, 

 SSRMEnable                 => SSRMEnable, 

 ParameterTransmissionReady => ParameterTransmissionReady, 

 DiagParameterError         => DiagParameterError 

 ); 

 

// Safe Speed Monitoring 0 

KFEI_SSM0( 

 DeactivateSS1M             := TRUE,  // SSM0 deactivated 

 Enable                     => SSM0Enable 

 ); 

 

// Safe Operating Stop Monitoring 

KFEI_SOSM( 

 DeactivateSS1M             := TRUE,  // SOSM deactivated 

 Set                        := SOSMSet, 

 Enable                     => SOSMEnable 

 ); 

 

// Safe Direction Monitoring 

KFEI_SDIM( 

 PosDeactivateSS1M          := FALSE,  // SDI pos. direction activated 

 NegDeactivateSS1M          := TRUE,  // SDI neg. direction deactivated 

 Set                        := SDIMSet, 

 PosEnable                  => PosEnable, 

 NegEnable                  => NegEnable 

 ); 

 

// Base functions of the module PSSu K F EI 

KFEI_Basic( 

 ActivateCascading          := FALSE, 

 SetO0                      := TimerEnableServo.Q, 

 SetO1                      := EnableOutputs, 

 SetO2                      := EnableOutputs, 

 ResetRestartInterlock      := ResetRestartInterlock, 

 ResetError                 := ResetError, 

 SpeedValid                 => SpeedValid, 

 SpeedValue                 => SpeedValue, 

 StateCascading             => StateCascading, 

 StateSTO_Timer1            => StateSTO_Timer1, 

 StateSTO_Timer2            => StateSTO_Timer2, 

 AutomaticSTO               => AutomaticSTO, 

 DiagRestartInterlock       => DiagRestartInterlock, 

 DiagEncoderMissing         => DiagEncoderMissing, 

 DiagModuleError            => DiagModuleError 

 ); 

 

// Switch-on delay for the HW enable of the servo amplifier 

TimerEnableServo( 

 IN := EnableOutputs AND NOT DiagRestartInterlock, 

 PT := T#200ms 

 ); 

 

//-- Messages --------------------------------------------------------------- 
SSM1Error_SR(   // Shutdown via SDIM 

SET1  := NOT SafetyGateEnable AND NOT TimerChangeSpeedRangeRunning AND 

         ParameterTransmissionReady AND NOT DiagParameterError AND NOT 

         SSM1Enable AND SpeedValid, 

 RESET := ResetRestartInterlock, 

 Q1    => SSM1Error 

 ); 

SDIMError_SR(   // Shutdown via SDIM 

 SET1  := NOT TimerChangeSpeedRangeRunning AND NOT PosEnable AND SpeedValid, 

 RESET := ResetRestartInterlock, 

 Q1    => SDIMError 

 ); 

END_PROGRAM 

3.5.1.3. Resource assignment 

 

Fig. 13: FS resource assignment 

3.5.2. I/O mapping 

The following I/O mappings are to be performed for the module PSSu K F EI (PI Variables <-> 

Module Bus): 

 

Fig. 14: I/O mappings of the module PSSu K F EI   

